

MOS Sleep Scale:

A Manual for Use and Scoring, Version 1.0

**Karen L. Spritzer and Ron D. Hays
November 2003**

The following citation is suggested when referencing this manual:

Spritzer, K. L. & Hays, R. D. (2003, November). MOS Sleep Scale: A Manual for Use and Scoring, Version 1.0. Los Angeles, CA.

SCORING RULES

Scoring the MOS Sleep Survey is a two-step process:

- First, original numeric values from the survey are recoded following the scoring rules outlined in Table 1. All items are scored so that a high score reflects more of the attribute implied by the scale name. Each item is then converted to a 0 to 100 possible range so that the lowest and highest possible scores are set at 0 and 100, respectively. In this format, scores represent the achieved percentage of the total possible score. For example, a score of 50 represents 50% of the highest possible score.
- Second, items within each scale are averaged together to create the 7 scale scores. Table 2a indicates which items contribute to each scale. Scales with at least one item answered can be used to generate a scale score. Items that are left blank (missing data) are not taken into account when calculating the scale scores. Scores represent the average for all items in the scale that the respondent answered.

Two additional measures are based on the average number of hours sleep each night during the past 4 weeks and are described in table 2b.

Table 1. Scoring Key: Recoding of Items

ITEM NUMBERS	Original response category	To recoded value of	
(not reversed) 4 ^(†) , 12 ^(†)	<1 ---->	.	
	1 ---->	0	
	2 ---->	20	
	3 ---->	40	bigger
	4 ---->	60	↑ problem
	5 ---->	80	
	6 ---->	100	
	>6 ---->	.	
(not reversed) 1	<1 ---->	.	
	1 ---->	0	
	2 ---->	25	bigger
	3 ---->	50	↓ problem
	4 ---->	75	
	5 ---->	100	
(reversed) 3, 5, 6, 7, 8, 9, 10, 11	<1 ---->	.	
	1 ---->	100	
	2 ---->	80	
	3 ---->	60	bigger
	4 ---->	40	↑ problem
	5 ---->	20	
	6 ---->	0	
	>6 ---->	.	
(reversed) 4 ^(†) , 12 ^(†)	<1 ---->	.	
	1 ---->	100	
	2 ---->	80	
	3 ---->	60	bigger
	4 ---->	40	↓ problem
	5 ---->	20	
	6 ---->	0	
	>6 ---->	.	

^(†)Note that items 4 and 12 are scored differently depending upon which scale they are used in. When used in the sleep adequacy scale, a higher score reflects less of a problem. When used in the sleep problems index, a higher score reflects more of a problem.

Table 2a: Averaging Items to Generate Sleep Scales

Scale	Number of Items	After Recoding Per Table 1, Average the Following Items
SLPD4 sleep disturbance	4	1, 3(R), 7(R), 8(R)
SLPSNR1 snoring	1	10(R)
SLPSOB1 sleep short of breath or headache	1	5(R)
SLPA2 sleep adequacy	2	4(R), 12(R)
SLPS3 sleep somnolence	3	6(R), 9(R), 11(R)
SLP6 sleep problems index I	6	4, 5(R), 7(R), 8(R), 9(R), 12
SLP9 sleep problems index II	9	1, 3(R), 4, 5(R), 6(R), 7(R), 8(R), 9(R), 12

(R) refers to a reversed item.

Table 2b: 3 Additional measures based on average number of hours sleep each night during the past 4 weeks

SLPQRAW Sleep Quantity (raw scores): use item 2 as is.

SLPQRAW=MOS2;

SLPOP1 Optimal Sleep (dichotomy):

MOS2	SLPOP1
<1	. (out-of-range)
1-6	0
7-8	1
9-23	0
24+	. (out-of-range)

Table 3: Central Tendency, variability (including floor and ceiling effects), and reliability of sleep scales[†]

Measure		Mean	Standard Deviation	% Floor	% Ceiling	Internal Consistency Reliability
SLPD4	sleep disturbance	29.20	23.37	8.56	0.56	0.84
SLPSNR1	snoring	30.89	30.25	32.52	6.00	NA
SLPSOB1	sleep short of breath or headache	13.29	21.82	64.23	0.87	NA
SLPA2	sleep adequacy	60.67	25.38	3.00	5.83	0.75
SLPS3	sleep somnolence	26.41	19.82	9.42	0.32	0.75
SLPQRAW	sleep quantity (raw)	6.93	1.40	0.06	0.03	NA
SLPOP1	optimal sleep (dichot)	0.54	0.50	45.58	54.42	NA
SLP6	sleep problems index I	28.31	18.09	2.32	0.03	0.78
SLP9	sleep problems index II	29.15	18.04	1.17	0.06	0.85

[†] Data is from the Medical Outcomes Study (MOS), n=3445.

NA - Not applicable for a single-item measure.

APPENDIX

- MOS Sleep Survey (see separate attachment).
- Crosswalk of item numbering between MOS Sleep survey and original Hays/Stewart paper:

MOS Sleep Survey	Hays/Stewart Paper
1	2
2	7
3	3
4	8
5	6
6	10
7	1
8	4
9	11
10	5
11	12
12	9

- Reference for Hays/Stewart paper:

Hays, R. D. & Stewart, A. L. (1992). Sleep measures. In A. L. Stewart & J. E. Ware (eds.), Measuring functioning and well-being: The Medical Outcomes Study approach (pp 235-259), Durham, NC: Duke University Press.

- SAS 8.02® Code for creating Sleep Scales:

```

*-----;
* sleep12-1.sas: ;
* ;
* SAS(tm) program to create scale scores and produce ;
* descriptive statistics for MOS Sleep Scale 1.0 ;
* measures. ;
* ;
* Written by Karen Spritzer, 11/25/2003. ;
*-----;
* ;
* This program assumes that your SLEEP items are in a ;
* dataset named "temp" and have variable names that ;
* correspond to the survey: MOS1-MOS12 ;
* and rescores them according to the rules found in the ;
* MOS Sleep Scale scoring manual: ;
* Spritzer, K. L. & Hays, R. D. (2003, November). MOS ;
* Sleep Scale: A Manual for Use and Scoring, Version 1.0. ;
* Los Angeles, CA. ;
* ;
* See http://gim.med.ucla.edu/FacultyPages/Hays/SLEEP.htm ;
* for scoring manual. ;
* ;
*-----;

*****;
data temp; set temp;

* recode out of range values for individual sleep items to missing;

IF MOS1 NOT IN(1,2,3,4,5) THEN MOS1=.;

ARRAY PT6 MOS3-MOS12;
DO OVER PT6;
IF PT6 NOT IN (1,2,3,4,5,6) THEN PT6=.;
END;

IF MOS2 < 1 OR MOS2 > 23 THEN MOS2=.;

/* COMPUTING NEW SLEEP SCALES */

* REVERSE ITEMS WHEN NECESSARY AND PUT ON 0-100 SCALE;

array orig1 MOS3 MOS4 MOS5 MOS6 MOS7 MOS8 MOS9 MOS10 MOS11 MOS12 ;
array rev1 rMOS3 rMOS4 rMOS5 rMOS6 rMOS7 rMOS8 rMOS9 rMOS10 rMOS11 rMOS12;
do over orig1;
rev1=(7-orig1-1)*20;
end;

MOS1 =(MOS1-1)*25;
MOS4 =(MOS4-1)*20;
MOS12=(MOS12-1)*20;

```

```

label
rMOS3      = "rev Q610A Feel that your sleep was not quiet"
rMOS4      = "rev Q610B Get enough sleep to feel rested up"
rMOS5      = "rev Q610C Awaken short of breath or with a h"
rMOS6      = "rev Q610D Feel drowsy or sleepy during the d"
rMOS7      = "rev Q610E Have trouble falling asleep - How "
rMOS8      = "rev Q610F Awaken during your sleep time and "
rMOS9      = "rev Q610G Have trouble staying awake during "
rMOS10     = "rev Q610H Snore during your sleep - How ofte"
rMOS11     = "rev Q610I Take naps of 5 minutes or longer d"
rMOS12     = "rev Q610J Get the amount of sleep you needed";

```

```
* CREATE SCALES by averaging items;
```

```
* Sleep Disturbance;
```

```
SLPD4 = mean(of MOS1, rMOS3, rMOS7, rMOS8);
label SLPD4 = "Sleep Disturbance Scale";
```

```
* Snoring;
```

```
SLPSNR1 = rMOS10;
label SLPSNR1 = "Snoring Scale";
```

```
* Short of Breath;
```

```
SLPSOB1 = rMOS5;
label SLPSOB1 = "Short of Breath Scale";
```

```
* Sleep Adequacy;
```

```
SLPA2 = mean(of rMOS4, rMOS12);
label SLPA2 = "Sleep Adequacy";
```

```
* Somnolence;
```

```
SLPS3 = mean(of rMOS6, rMOS9, rMOS11);
label SLPS3 = "Somnolence Scale";
```

```
* Sleep Problems Index I;
```

```
SLP6 = mean(of MOS4, rMOS5, rMOS7, rMOS8, rMOS9, MOS12);
label SLP6 = "Sleep Problems Index I";
```

```
* Sleep Problems Index II;
```

```
SLP9 = mean(of MOS1, rMOS3, MOS4, rMOS5, rMOS6, rMOS7, rMOS8, rMOS9, MOS12);
label SLP9 = "Sleep Problems Index II";
```

```

* Sleep Quantity (raw score)";
SLPQRAW=MOS2;
label SLPQRAW = "Sleep Quantity (raw score)";

* Optimal Sleep;
if MOS2>=0 then do;
if (MOS2 = 7 or MOS2 = 8) then SLPOP1=1; else
if (MOS2 < 7 or MOS2 > 8) then SLPOP1=0;
end;
label SLPOP1 = "Optimal Sleep Scale";

run;
*****;

TITLE1 "Descriptive statistics on MOS SLEEP scales"; run;

proc means maxdec=2 data=temp;
var SLPD4 SLPSNR1 SLPSOB1 SLPA2 SLPS3 SLP6 SLP9 slpgraw slpop1;
run;

TITLE1 "Alpha reliability for MOS SLEEP scales"; RUN;

TITLE2 "SLPD4 Sleep Disturbance";
proc corr alpha nomiss;
var MOS1 rMOS3 rMOS7 rMOS8;
run;

TITLE2 "SLPA2 Sleep Adequacy";
proc corr alpha nomiss;
var rMOS4 rMOS12;
run;

TITLE2 "SLPS3 Somnolence";
proc corr alpha nomiss;
var rMOS6 rMOS9 rMOS11;
run;

TITLE2 "SLP6 Sleep Problems Index I";
proc corr alpha nomiss;
var MOS4 rMOS5 rMOS7 rMOS8 rMOS9 MOS12;
run;

TITLE2 "SLP9 Sleep Problems Index II";
proc corr alpha nomiss;
var MOS1 rMOS3 MOS4 rMOS5 rMOS6 rMOS7 rMOS8 rMOS9 MOS12;
run;

```